

10.1 被験者間 2 要因分散分析 (完全無作為要因計画)

従属変数に対して影響をもつと考えられる要因が 2 つ以上ある場合にも、平均値を比較する場合に分散分析を適用することができる。2 要因分散分析には、それぞれの要因が被験者間であるか被験者内であるかによって 3 つのパターン(要因計画)があり、それぞれ

- 1) 被験者間 2 要因(完全無作為要因計画)
- 2) 被験者内 2 要因(乱塊法計画)
- 3) 被験者内×被験者間 2 要因(分割法計画)

と呼ぶ。ここでは 1) についてのみ解説する。

ある従属変数に対して影響を及ぼしているかどうかを調べたい要因が A と B の 2 つあり、これらの要因に実験参加者を完全にランダムに割り付ける計画を「完全無作為要因計画(Completely Randomized Factorial Design)」という。この場合、実験参加者は、いくつかある実験状況のうち、自分が割り付けられた 1 つの下でしか課題を遂行しない。よって要因はそれぞれ 対応がない = 被験者間要因 であるといえる。その中でも、各要因に 2 水準が設定されている、もっとも単純なケースを考えてみよう。

たとえば、向後 Web 教材 (<http://www.kogolab.jp/elearn/hamburger/chap7/>) に示されたフライドチキンの例を考えてみよう。食感(クリスピーか普通の衣か)と味付け(辛口と普通味)の異なる 4 種類のフライドチキンのうち、どれがより「おいしい」と好まれるかについて調べるために、15 名ずつの街の人に 1 種類ずつを食べてもらい、100 点満点で点数をつけてもらった結果が下表に示したとおりである。こうした実験方法が「完全無作為要因計画」にあたる。

クリスピー		普通の衣	
辛口	普通味	辛口	普通味
65	65	70	70
85	70	65	70
75	80	85	85
85	75	80	80
75	70	75	65
80	60	65	75
90	65	75	65
75	70	60	85
85	85	85	80
65	60	65	60
75	65	75	70
85	75	70	75
80	70	65	70
85	80	80	80
90	75	75	85

2 要因分散分析では、以下のような検討をおこなう。1 要因分散分析と異なるのは、要因 A と B それぞれの効果(主効果)と同時に、それらの組み合わせによって生じる(裏返して言えば、どちらか一方の要因を操作しただけでは出現しえないような)効果、すなわち交互作用効果の検討が加わる点である。

1) 主効果の検討

食感 and/or 味付けの有意な主効果があるか?

2) 交互作用の検討

食感×味付けの有意な交互作用効果があるか? : 単純主効果の検定(食感別の味付け効果の検討)

2 要因分散分析の場合, 分散分析の時に考えなければならないデータのズレについて考えるときに, 2 つ(以上)の要因の組み合わせによって生じる効果, すなわち「交互作用」が存在することを考慮しなければならないことが 1 要因分散分析の場合とは異なっている.

2 要因分散分析の場合, データ全体におけるズレはどのように分解できるかという,

$$\text{全体のズレ} = \text{要因 1 によるズレ} + \text{要因 2 によるズレ} + \text{交互作用によるズレ} + \text{偶然によるズレ}$$

となる. 要因 1 によるズレ, 要因 2 によるズレ, そして交互作用によるズレのそれぞれを偶然によるズレと比べて, 各効果が有意であるかどうかを検討することになる. また, 偶然のズレのことを一般に「残差」とよぶ.

それでは, SAS による分析をおこなってみよう. 基本形は, 1 要因分散分析と同じである. ここでは, do ループを使ったデータ読み込みパターンを用いたプログラムを組んでみる. ただし, 被験者間要因の分散分析の場合, データの入力の形は(ある被験者は 1 つの条件しか遂行しないので)プログラムに影響しない.

```
data crf24;
  do koromo=1 to 2;                /* 食感がクリスピー(1)か普通衣(2)か */
    do taste=1 to 2;              /* 味付けがスパイシー(1)か普通味(2)か */
      input point @@; output;     /* 食べた人がつけた点数を読み込み */
    end; end;

cards;
65 65 70 70
85 70 65 70
75 80 85 85
85 75 80 80
75 70 75 65
80 60 65 75
90 65 75 65
75 70 60 85
85 85 85 80
65 60 65 60
75 65 75 70
85 75 70 75
80 70 65 70
85 80 80 80
90 75 75 85
;
proc glm;
class koromo taste;
model point=koromo taste koromo*taste /ss3;      /*検定する効果を並べ, 平方和を指定*/
means koromo taste /tukey;                       /*主効果の多重比較*/
means koromo*taste;                              /*各 4 条件での平均値の出力*/
run;
quit;
```

```

data crf24;
  input sub koromo taste point;

cards;
1 1 1 65
2 1 2 65
3 2 1 70
4 2 2 70
...
57 1 1 90
58 1 2 75
59 2 1 75
60 2 1 85
;

```

注:

上記の例では do ループを用いてデータを読み込んでいるが、1 行に実験参加者 1 名のデータを記述する左記のスタイルであっても、プロセスステップの記述は変わらない(つまり、被験者間要因のみを扱う限り、データセットの並びがプログラムに影響を及ぼすことはない)。この場合は、データの並び順が input 文のようになり、左のようなデータセットになることをよく理解すること。

output の見方

主効果と交互作用の検定

Dependent Variable: point

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	3	634.583333	211.527778	3.52	0.0209
Error	56	3370.000000	60.178571		
Corrected Total	59	4004.583333			

R-Square	Coeff Var	Root MSE	point Mean
0.158464	10.42439	7.757485	74.41667

Source	DF	Type III SS	Mean Square	F Value	Pr > F
koromo	1	50.4166667	50.4166667	0.84	<u>0.3640</u>
taste	1	183.7500000	183.7500000	3.05	<u>0.0860</u>
koromo*taste	1	400.4166667	400.4166667	6.65	<u>0.0120</u>

四角で囲んだ数値が、食感と味付けそれぞれの主効果、そして両者の交互作用に関する有意性の検定結果である。この結果から、食感と味付けの主効果はいずれも有意ではなかったが、食感×味付けの交互作用が有意であることがわかる。食感と味付けはそれぞれ単独では「おいしさ」を決める要因とはなりえていないが、両者が組み合わさることによって「おいしさ」の評価に違いが生じている異なる可能性が示されたわけだ。では、どのような組み合わせによるどのような「おいしさ」の違いがあるのだろうか？ 各群の平均値は means koromo*taste; によって出力されており、

Level of koromo	Level of taste	N	Mean	Std Dev
1	1	15	79.6666667	7.89816133
1	2	15	71.0000000	7.36788398
2	1	15	72.6666667	7.76132046
2	2	15	74.3333333	7.98808637

となっている。平均値を「眺める」と、なんとなくどういう効果が見当をつけることはできるだろう。

さて、2 要因の分散分析の場合、ここまでの出力結果から、次に進むべき道が、以下の 3 つの選択肢のうちのいずれか 1 つに決まる

- 1) 交互作用が見られた場合: 単純主効果の検定と多重比較へ(主効果の多重比較はおこなわない)
- 2) 主効果が見られ、交互作用が見られなかった場合: 主効果の多重比較へ
- 3) 主効果も交互作用も見られなかった場合: 残念!!(ここで終了)

ここでは 1)の道を選ぶことになる。もし 2), すなわち主効果のいずれか(も)が有意で、交互作用が有意でなかった場合は、`means koromo taste /tukey;`により実行される主効果の多重比較の出力結果のうち、有意な効果が得られた方を参照する。以下の手続きは 1 要因分散分析の場合と同じなので、繰り返す必要はないだろう。¹ このように、ある要因の主効果のみが見られ、交互作用が見られなかった場合は、この多重比較結果を記述して分析終了となる。

ではここでは 1)に戻ることにしよう。食感×味付けの交互作用が見られたということは、食感によって味付けの効果が異なる(あるいは味付けによって食感の効果が異なる)らしいということが予想される。具体的にはどのように「異なる」のだろうか。これを読み解くためにおこなうのが、`単純主効果の検定`である。単純主効果の検定とは「食感ごとの味付けの効果」「味付けごとの食感の効果」というように、片方の要因をコントロールした状況下で、もう片方の要因の効果がどのようなものであるかを検討することである。ここでは「食感ごとの味付けの効果」を見てみることにする。

この単純主効果の検定をおこなうためには、分散分析のプロシジャに以下の記述を追加する。

```
lsmeans koromo*taste /slice=koromo;
```

lsmeans ステートメントは、単純主効果を検討するためのステートメントであり、/slice オプションで、どの要因ごとの検討をおこなうかを指定する。本来 lsmeans とは各要因の水準ごとの従属変数の`調整済み平均`(すなわち各水準でデータ数が異なる場合に、そのデータ数の違いを調整した平均値)を計算するオプションである。ただし、単純主効果を求めるために/slice オプションを指定する場合は、各水準でデータ数が同じでも違っててもこのステートメントを用いなければならない。means ステートメントに/slice オプションをつけてもエラーが出る。

交互作用の単純主効果の検定

Least Squares Means

koromo*taste Effect Sliced by koromo for point

koromo	DF	Sum of Squares	Mean Square	F Value	Pr > F
1	1	563.333333	563.333333	9.36	0.0034
2	1	20.833333	20.833333	0.35	0.5586

この結果から、koromo=1 すなわちクリスマスケーキの場合に、味付けの単純主効果が有意であることが分かる。一方で、koromo=2 すなわち普通衣の場合には、味付けの単純主効果は有意ではない。そこで、先ほどの各群における「おいしさ」評価の平均値に立ち返ってみると、クリスマスケーキの場合には辛口が普通味よりぐんと点数が高いのに対して、普通衣の場合には辛口と普通味の点数に違いがないということになる。どうやら、4 種のチキンのうち、クリスマスケーキ×辛口という組み合わせのチキンの評判が高いようである。²³ これを重点的に売り出せば、売上金額が伸びるのではないだろうか。

¹ ここではどちらの主効果も有意ではなく、当然 Tukey 法による多重比較の結果にも有意差はない

² Web 教材では、交互作用について厳密な統計的検定をおこなっていないため、解釈があいまいになっている

³ 味付けの要因に 3 水準(例:スパイシー味・チーズ味・普通味)以上ある場合は、単純主効果の多重比較も必要になるが、ここでは複雑になりすぎるのでふれない